
Executive Summary 
 
 The purpose project was to develop a software application to determine various 

economic parameters including present value, net present value, as well as discount cash 

flow rate of return.  We were given the cash flow, and the investment, both monetary, and 

equipment.  Using this data, a program was written to calculate these necessary 

parameters.  The user can input an interest rate, and increment this interest rate by fixed 

amounts so that a present value profile can be generated.  The present value profile is also 

generated by the program.  This program is also capable of calculating the DCFROR, an 

important indicator, which lets you determine the actual economic limitations of your 

project. 

Introduction 

 The cash flow of the project was provided as follows: 

Year Investment(M$) 
Revenue 

(M$) 
Operating 
Cost (M$) 

0 -1,000 --- --- 
1  540 $100 
2  400 $83 
3  380 $80 
4  350 $63 
5  300 $59 
6  200 $50 
7  185 $50 
8  160 $40 
9  100 $30 

10  56 $10 
Total  2,671  

  

The initial $1,000,000 investment includes all necessary elements to have a producing 

well in the oil industry.  Land, equipment (pumping unit at $250,000 initial cost with 6 

year life, as well as a $15,000 salvage value), drilling, and completion costs are all 

included in this investment..  The tax rate used in the calculations was 46%.  The 

program is not limited, however to this example cash flow.  It will calculate PV, NPV, 

and DCFROR for any cash flow provided it up to 10 years. 

Background 

Several yardsticks are used to measure the profitability of a project.  These 

yardstick are also used in mutually exclusive decisions.  Net present value is one of these 

yardsticks.  It shows the value of the project in todays dollars, which is a way to 

normalize projects of different costs, and durations to maximize the potential for profit. 

Discount cash flow rate of return is also a good economic indicator.  It is the 

interest rate at which the project breaks even, or in other words, makes no money.  This is 

valuable, because when you find the DCFROR, you will know the highest possible 

interest rate that you can borrow money, and still make a profit.  For some oil companies, 

a DCFROR or 25% or higher is desirable. 

However useful these indicators are, they are useless without production data, 

offset costs, as well as nominal prices of hydrocarbons.  For these sets of data, many 

things can be used, offset well data, and decline curve analysis may be used to predict 

well performance.  Calculating offset, and operational costs is also necessary, but is 

beyond the scope of this project.  As it is impossible to predict the price of oil in the 

future, an educated informed, and reasonable nominal price should be chosen.  Preferably 

a good nominal price should be one that is inline with historical pricing data. 

Problem Statement 

Given a cash flow, we were to write a computer program that calculates net 

present value, discount cash flow rate of return, as well as making a present value profile 

for that cash flow. 

Methodology 

 First, you must make an analysis of the cash flow to determine taxes, and hence 

the present value of the project at that given year.  These present values are summed to 

yield the total net present value of the project.   

Year 
Investment 

(M$) 
Revenue 

(M$) 
Operating 
Cost (M$) 

Depreciation 
(M$) 

Income 
Tax 

($M) 
NCF (M$) NPV (M$) 

0 -$1,000     -$1,000 -$1,000 
1  $540 $100 $41.67 $183.23 $256.77 $238.85 
2  $400 $83 $41.67 $126.65 $190.35 $164.71 
3  $380 $80 $41.67 $118.83 $181.17 $145.83 
4  $350 $63 $41.67 $112.85 $174.15 $130.40 
5  $300 $59 $41.67 $91.69 $149.31 $104.00 
6  $200 $50 $26.67 $56.73 $93.27 $60.43 
7  $185 $50  $62.1 $72.9 $43.94 
8  $160 $40  $55.2 $64.8 $36.33 
9  $100 $30  $32.2 $37.8 $19.72 
10  $56 $10  $21.16 $24.84 $12.05 

Total  $2,671    $245.34 -$43.73 

 

NOTE:  Straight line depreciation was used for depreciating the pump in this example. 

 

Equations utilized were: 

[ ] %46Re ×−−= onDepreciatiCostOperatingvenueTaxIncome  

TaxIncomeCostOperatingvenuewNetCashFlo −−= Re  

( )ni
NCF

Valueesent
+

=
1

*where i is the interest rate, and n is the number of years .   

These present values are then summed to calculate the NPV.  To calculate the DCFROR 

of this project, a do until loop was utilized to iterate on interest rate until the NPV of the 

project was 1 dollar. 

Procedure 

Net cash flow was calculated by importing a text file that was prepared previously 

to running the program by the user.  These values are read into the program, andby the 

simple cluck of a button, the present value, and hence the net present values of the project 

were found.  The default interest rate of this program is 7.5%, however, it can be an 

arbitrary rate, decided by the user.   

The present value profile is generated by clicking on the graph tab of the program.  

The user is then able to choose what parameters he would like to graph versus other 

parameters.  For present value profile, NPV (on the Y axis) versus i (on the x axis) must 

be selected via drop down menus.  If the wrong combinations of parameters are chosen, 

the program will guide the user on how to correctly use the graphs. The user is also able 

to enter a beginning interest rate, an increment, as well as the number of points they 

desire  After the correct parameters are chosen by the user, simply click the graph button, 

and the PVP will be generated.   

The DCFROR in this program is not graphed, however, it is calculated, and 

displayed via clicking the calculate the DCFROR button on the main tab of the program. 

Results 

 From the previous table which was shown in the methodology section of this 

report, a NPV was calculated to be a negative 43,724 dollars.  This particular cash flow 

also had a DCFROR of about 6.14%.  Please see following screen prints to view 

program. 

  

 
Figure 1 – User’s ability to open files containing data 

 
 
Figure 2- opening the pre-made NCF file 

Figure 3 – PV, NPV, and DCFROR results printed to screen 

 
 
Figure 4 – NPV calculated at different interest rates. 

Figure 5 – Cash Flow Diagram  
 

 
 
Figure 6 – Present Value Profile.  Note DCFROR at 6.14% 

Final Conclusions 

From hand calculations, and comparison to the answers yielded by the program, I  
 
conclude that this program is accurate in calculating both DCFROR as well as present  
 
value, as well as net present value. This program was also tested against examples in the  
 
text, and was also correct.  Hence, This program will work properly given any cash flow 
 
provided by the user. 

Appendix 
 
Source Code 
 
Option Explicit 
Dim NCF() As Single 
Dim NumberofCases As Integer 
 
Private Sub cmdDCFROR_Click()     
    Dim i As Integer 
    Dim AIR As Single 
    Dim NCF As Single 
    Dim Year As Integer 
    Dim pv As Single 
    Dim NPV As Single 
     
    AIR = 0 
    NPV = 2 
     
    Do Until NPV < 1         
     AIR = AIR + 0.0001 
     NPV = 0 
         
        With grdMain             
             For i = 1 To NumberofCases                 
                .Row = i                 
                .Col = 0 
                Year = Val(.Text) 
             
                .Col = 1 
                 NCF = Val(.Text)             
                pv = NCF / (1 + AIR) ^ Year 
                NPV = NPV + pv 
             
            Next i             
        End With         
    Loop 
     
    txtDCFROR.Text = Round(AIR * 100, 2) 
         
End Sub 
 
Private Sub cmdSaveNPV_Click() 
     
    Dim NoOfpoints As Integer 
    Dim NPVFile As String 

    Dim i As Integer     
    NoOfpoints = Val(txtPointAIR.Text)     
    With cdgOpenSaveFile         
         
        .DialogTitle = "Save" 
        .CancelError = False 
        .Filter = "Text File|*.txt|All Files (*.*)|*.*" 
        .ShowSave         
        If Len(.FileName) = 0 Then Exit Sub             
            NPVFile = .FileName 
             
    End With 
     
    Open NPVFile For Output As #1 
         
       With grdNPV 
             For i = 0 To NoOfpoints                 
            Print #1, .TextMatrix(i, 0), .TextMatrix(i, 1)             
            Next i         
        End With 
     
    Close #1 
     
End Sub 
 
Private Sub Form_Load() 
        
    Call PrepareGrid 
    Call PrepareNPVGrid 
     
End Sub 
 
Private Sub PrepareGrid() 
     
    With grdMain 
         
        .Row = 0 
        .Col = 0 
        .CellAlignment = 4 
        .CellFontBold = True 
        .Text = "Year" 
     
        .Row = 0 
        .Col = 1 
        .CellAlignment = 4 
        .CellFontBold = True 

        .Text = "NCF" 
     
        .Row = 0 
        .Col = 2 
        .CellAlignment = 4 
        .CellFontBold = True 
        .Text = "PV" 
         
    End With 
     
End Sub 
 
Private Sub PutNcfInGrid() 
     
    Dim i As Integer 
     
    With grdMain         
        .Rows = NumberofCases + 1         
        For i = 1 To NumberofCases 
             
         .Row = i 
         .Col = 0 
         .CellAlignment = 4 
         .Text = i - 1 
             
         .Col = 1 
         .CellAlignment = 4 
         .Text = NCF(i)             
        Next i 
     
      End With 
     
End Sub 
 
Private Sub cmdPV_Click() 
     
    Dim i As Integer 
    Dim MyAIR As Single 
    Dim MyNCF As Single 
    Dim MyYear As Integer 
    Dim pv As Single 
     
    MyAIR = 0.01 * Val(txtAIR.Text) 
     
    With grdMain         
        For i = 1 To NumberofCases             

            .Row = i             
            .Col = 0 
            MyYear = Val(.Text)             
            .Col = 1 
            MyNCF = Val(.Text)             
            pv = MyNCF / (1 + MyAIR) ^ MyYear             
            .Col = 2 
            .CellAlignment = 4 
            .Text = Round(pv, 2)             
        Next i 
         
    End With 
     
End Sub 
 
Private Sub cmdNPV_Click() 
     
    Dim MyAIR As Single     
    MyAIR = Val(txtAIR.Text)     
    txtNPV.Text = Round(NetPresentValue(MyAIR), 2) 
     
End Sub 
 
Private Function NetPresentValue(AIR As Single) As Double     
    Dim i As Integer 
    Dim pv As Single 
    Dim MyYear As Integer 
    Dim MyNCF As Single     
    NetPresentValue = 0     
    
 With grdMain         
        For i = 1 To NumberofCases             
            MyYear = Val(.TextMatrix(i, 0)) 
            MyNCF = Val(.TextMatrix(i, 1)) 
            pv = MyNCF / (1 + AIR / 100) ^ MyYear 
            NetPresentValue = NetPresentValue + pv             
        Next i         
    End With     
End Function 
 
Private Sub cmdDrawGraph_Click() 
     
    Dim i As Integer 
    Dim j As Integer 
    Dim NoDataSets As Integer 

    If cmbYAxis.Text = "NPV" Then 
            If cmbXAxis.Text <> "AIR" Then 
                i = MsgBox("The X axis should be AIR", vbOKOnly, "Error") 
                Exit Sub 
            End If 
            NoDataSets = Val(txtPointAIR.Text) 
        Else 
             If cmbXAxis.Text <> "Year" Then 
                i = MsgBox("The X axis should be Year", vbOKOnly, "Error") 
                Exit Sub 
            End If 
            NoDataSets = NumberofCases 
    End If 
        
    If cmbYAxis.Text = "NPV" Then 
            NoDataSets = Val(txtPointAIR.Text) 
        Else 
            NoDataSets = NumberofCases 
    End If 
     
    ReDim XAxis(NoDataSets) As Single 
    ReDim YAxis(NoDataSets) As Single     
     
    For i = 1 To NoDataSets 
        Select Case cmbXAxis.Text 
            Case "Year" 
                XAxis(i) = Val(grdMain.TextMatrix(i, 0)) 
            Case "AIR" 
                XAxis(i) = Val(grdNPV.TextMatrix(i, 0)) 
        End Select 
            Select Case cmbYAxis.Text 
                Case "NCF" 
                    YAxis(i) = Val(grdMain.TextMatrix(i, 1)) 
                Case "PV" 
                    YAxis(i) = Val(grdMain.TextMatrix(i, 2)) 
                Case "NPV" 
                    YAxis(i) = Val(grdNPV.TextMatrix(i, 1)) 
            End Select 
    Next i 
    
    With chrMain         
        If optBarGraph.Value = True Then 
           .chartType = VtChChartType2dBar 
        Else 
            .chartType = VtChChartType2dLine 
        End If 

        .ColumnCount = 1 
        .RowCount = NoDataSets 
         
        For i = 1 To NoDataSets             
            .Row = i 
            .RowLabel = XAxis(i) 
            .Data = YAxis(i)             
        Next i 
         
    End With 
     
End Sub 
 
Private Sub PrepareNPVGrid() 
      
     With grdNPV 
         
        .Row = 0 
        .Col = 0 
        .ColAlignment(0) = 4 
        .CellFontBold = True 
        .Text = "AIR" 
     
        .Row = 0 
        .Col = 1 
        .ColAlignment(1) = 4 
        .CellFontBold = True 
        .Text = "NPV" 
     
    End With 
     
End Sub 
 
Private Sub cmdCalculateNPVGrid_Click() 
     
    Dim i As Integer 
    Dim AIR As Single 
    Dim Increment As Single 
    Dim NoOfpoints As Integer 
     
    AIR = Val(txtStart.Text) 
    Increment = Val(txtIncrement.Text) 
    NoOfpoints = Val(txtPointAIR.Text) 
     
    With grdNPV 

        .Rows = NoOfpoints + 1         
        For i = 1 To NoOfpoints            
           .TextMatrix(i, 0) = AIR 
           .TextMatrix(i, 1) = Round(NetPresentValue(AIR), 2) 
           AIR = AIR + Increment 
         
        Next i 
     
    End With 
     
End Sub 
 
Private Sub mnuExit_Click() 
     
    End 
     
End Sub 
 
Private Sub mnuNcfFile_Click() 
     
    Dim NCFFile As String 
    Dim Count As Integer 
     
    Count = 0 
     
    With cdgOpenSaveFile         
        .DialogTitle = "Open" 
        .CancelError = False 
        .Filter = "NCF File|*.txt|AllFiles(*.*)|*.*" 
        .ShowOpen         
        If Len(.FileName) = 0 Then Exit Sub         
        NCFFile = .FileName 
     
    End With 
     
    Open NCFFile For Input As #1 
         
        Do Until EOF(1)             
            Count = Count + 1             
            ReDim Preserve NCF(Count)             
            Input #1, NCF(Count)         
        Loop 
     
    Close #1     
    NumberofCases = Count     

    Call PutNcfInGrid 
     
End Sub 
 
Private Sub mnuNPV_Click() 
     
    Call cmdSaveNPV_Click 
     
End Sub 

